

## Un générateur de bruit rose

*J'invite le lecteur à consulter le site pour des informations complémentaires.*

Page d'accueil du site Internet :

[page d'accueil](#)

d'autres pdf, sur différents sujets :


[liste des PDF](#)

*Le bruit rose est utilisé en acoustique, comme, par exemple, pour mesurer des caractéristiques des locaux sonorisés par des haut-parleurs. Il est également connu comme son ambiant pour apaiser et faciliter l'endormissement. Il existe donc des générateurs de bruit rose. Le noyau de base est un bruit blanc, et on ajoute un filtre particulier pour réaliser un bruit rose.*

### 1) La constitution d'un générateur de bruit

#### 1.a. Bruit blanc

Le bruit blanc est un signal qui contient de façon identique toutes les fréquences, sur une large étendue. (On peut faire l'analogie avec la lumière blanche qui contient toutes les couleurs du spectre visible).




*profil de puissance d'un bruit blanc*

Pour réaliser un bruit blanc, il existe 2 principes très différents :

#### - bruit blanc analogique

##### Schéma


On exploite le bruit électronique qui prend naissance dans une jonction polarisée en inverse. Cette jonction peut être celle de Base Emetteur d'un transistor à collecteur ouvert, ou une diode zener polarisée à proximité de sa tension de claquage. La tension recueillie n'est que de quelques millivolts efficace, ce qui justifie l'emploi d'un amplificateur de tension en aval.



*exemple de schéma pour générer du bruit blanc*

##### Spectre

Le signal électronique est aléatoire, lié à des phénomènes qui se créent au sein de la zone de charge d'espace, à échelle microscopique, notamment le bruit de « grenaille » qui peut être vu comme une suite aléatoire d'impulsions de Dirac (intervalles irréguliers, « amplitudes » irrégulières). Ce signal a une densité spectrale de puissance constante.



*aspects temporel et fréquentiel*


Le bruit thermique dû à la résistance ajoute un bruit blanc non corrélé.

- bruit blanc numérique

Schéma

On exploite une succession de 0 et 1 qui se présente en sortie d'un registre à décalage, cadencé par une horloge, associé à des portes XOR.

Le signal est en fait pseudo aléatoire, c'est à dire qu'il existe un (très long) motif qui se reproduit. Cette structure porte le nom de Pseudo Random Pattern Generator, PRPG.




exemple de schéma de PRPG

Spectre

Il est formé de raies, car le signal est périodique. Les raies sont espacées de  $F_{CLK}/N$ . La raie en 0 Hz est la valeur moyenne du signal.

Une fonction en  $\frac{\sin U}{U}$ , avec  $U = \pi \frac{f}{f_{CLK}}$

module l'amplitude des autres raies.



aspects temporel (ici 0 1 1 0 1 0 1 1 1 1 0 0 0 1 0) et fréquentiel (l'enveloppe en sinc passe par 0 à  $F_{CLK}$ )

Un générateur de bruit peut donc être réalisé à base de PRPG avec  $N \rightarrow \infty$ , et suivi d'un filtre passe bas, coupant à  $F_{CLK}/20$ . De cette façon :

- les raies sont très proches entre elles, car  $F_{CLK}/N \rightarrow 0$ ,

- les énergies de ces raies sont identiques à 1% près car, jusqu'à  $F_{CLK}/20$ , on a :  $1 > (\frac{\sin U}{U})^2 > 0,99$ .

Comparaison


Le montage analogique est très simple, mais il est difficile d'obtenir une densité spectrale de bruit plate. De plus, c'est tributaire des conditions de polarisation, et fatalement, de la température. Ce type de montage peut convenir dans des applications peu exigeantes.

La version numérique présente l'avantage d'une robustesse et d'une reproductibilité. 2 remarques :

- le fait d'avoir un spectre formé de raies n'est pas un inconvénient si le nombre N est très élevé,
- pour que le bruit soit blanc dans une large étendue de fréquence, il faut une fréquence d'horloge élevée.


1.b. Bruit rose

Quelle que soit la source de bruit blanc, on fabrique du bruit rose à l'aide d'un filtre particulier.



Il doit être du type passe bas, tel que sa réponse harmonique présente une pente de **- 10 dB / décade**. Dit autrement, sa densité spectrale de puissance est divisée par 10 quand la fréquence est multipliée par 10. Cela correspond à la loi en « 1/f ». Cette propriété de **puissance** de bruit par décade constante est exploitée en acoustique. Rappelons qu'une fonction de transfert en **puissance** s'écrit  $10 \log (P_s/P_e)$ .

La fonction de transfert en **tension**, soit  $20 \log (V_s/V_e)$ , présentera une pente de **- 10 dB / décade**, ou **3 dB / octave**.



réponse harmonique du filtre

A titre de renseignement, on peut écouter ces 2 bruits ici : [présentation](#)

## 2) Le filtre à - 10 dB/décade

### 2.a. Schéma

Une façon simple de faire ce filtre est analogique. Le schéma est donné ci-contre.

L'entrée est le nœud 1.  
La sortie est le nœud 100.




Schéma du filtre à 10 dB / décade

Il est issu du schéma interne du générateur de bruit PNG11, présenté en annexe 1. Quelques valeurs numériques ont été changées.

Il est constitué d'un réseau RC, et d'un amplificateur non inverseur.

La liaison entre ces 2 étages se fait par un condensateur ( $CL = 100 \mu F$ ) qui coupe la composante continue. La résistance  $R9 = 150 k\Omega$  ramène un potentiel statique de 0 V sur l'entrée +. Pour réduire un offset créé par IBIAS de l'amplificateur opérationnel, on retrouve  $R10 = 150 k\Omega$  sur l'entrée -.

L'amplification statique de l'étage à A. Op. est  $(1+Ra/Rb) = (1+18k/2,2k) = 9,18$ , soit un gain de 19 dB.

La structure du réseau RC est telle qu'il y a un rapport d'environ 4,7 entre deux branches consécutives:

	10 kΩ	4,7 kΩ	2,2 kΩ	1 kΩ	470 Ω	220 Ω	82 Ω	33 Ω
	2,2 μF	1 μF	470 nF	220 nF	100 nF	47 nF	22 nF	10 nF
RC	22 ms	4,7 ms	1,034 ms	0,22 ms	47 μs	10,34 μs	1,804 μs	0,33 μs
1/2πRC	7,2 Hz	34 Hz	154 Hz	720 Hz	3,4 kHz	15,4 kHz	88,4 kHz	482 kHz

Les fréquences de coupures associées couvrent un large spectre de fréquence.

### 2.b. Réponse harmonique en tension (netliste en annexe 2 : filtre.cir)

Une simulation .AC permet d'obtenir la réponse  $20 \log (V(100)/V(1))$ , où V(1) est l'entrée du filtre



Analyse harmonique en tension du filtre

#### Interprétation

On lit clairement que la pente à - 10 dB / décade est assurée de quelques Hz à 1 MHz environ. Au-delà, on est limité par la réponse de l'amplificateur opérationnel.

Le gain en basse fréquence est de 19 dB, comme déterminé. (En dessous de 0,1 Hz, la réponse en fréquence chute, car la composante continue est coupée).


Regardons 2 fréquences particulières : à 100 Hz, on dispose d'une légère amplification (1,1983) et à 10 kHz, l'atténuation en tension est de 0,125.

X Values	100.017	10.003K
V(100)	1.1983	125.002m

### 2.c. Simulation en temporel, signal d'entrée composite

Une façon de vérifier la réponse harmonique précédente est de placer en entrée plusieurs sinusoïdes de même amplitude (par exemple 1 V arbitraire), mais de fréquence différente. Pour faire le test sur un seul run, on fait la somme de 3 fréquences : 100 Hz, 1 kHz, 10 kHz, ce qui se situe au centre (sur l'échelle log) de la gamme à - 10 dB/décade.


En simulation, il suffit de placer les 3 sources en série.



Montage simulé

Remarque : Pour faire ce run, il suffit de commenter / décommenter les lignes dans la netliste filtre.cir.

#### Aspect temporel



Entrée et sortie du filtre, réponse temporelle


#### Interprétation

A l'échelle présentée (20 ms), on reconnaît clairement, sur le signal d'entrée, 2 périodes de 100 Hz, avec une superposition d'une composante à 1 kHz. Il faut zoomer en abscisses pour percevoir le signal 10 kHz. L'action du filtre privilégie la basse fréquence et réduit la fréquence 1 kHz.

#### Aspect fréquentiel

Par FFT, on visualise :  
le signal d'entrée, V(1) : 3 composantes fréquentielles ;  
et le signal de sortie V(100) :

X Values	100.000
V(100)	1.2037
X Values	1.0000K
V(100)	386.410m
X Values	10.000K
V(100)	124.651m



Entrée et sortie du filtre, réponse fréquentielle par FFT

#### Interprétation

La représentation fréquentielle montre bien 1 V d'amplitude pour les 3 raies du signal d'entrée, et respectivement 1,2 V, 386,4 mV, 124,6 mV pour la sortie.

On retrouve les amplifications en tension de la réponse harmonique d'une part, et on confirme la pente à - 10 dB/décade : sur 2 décades (100 Hz, 10 kHz), on a une atténuation de 20 dB.

### 3) Simulation de la source de bruit blanc

#### 3.a. Simuler un bruit blanc analogique

Pour confirmer le comportement de ce filtre face à une source de bruit blanc, on lance une simulation « NOISE », en réalisant tout d'abord une source de bruit parfaite.

Conformément à la théorie, Pspice utilise la densité spectrale de tension de bruit générée par une résistance R par la relation :  $\gamma = 4 k T R$ . (en  $V^2/Hz$ )

Il s'agit du bruit thermique, appelé aussi bruit Johnson.

C'est une densité spectrale indépendante de la fréquence, ce qui va nous être utile pour générer un bruit blanc pour la simulation.





schéma en bruit d'une résistance


Prenons une résistance placée en parallèle avec une source de tension nulle. Et choisissons la valeur de cette résistance à 4 k T.

Le courant de bruit circulant sera  $\gamma / R$ , soit  $4 k T R / 4 k T R = 1 A^2/Hz$ .



Ajoutons une source de tension commandée par ce courant : on a fabriqué un dipôle qui délivre une tension de bruit unitaire.

Pour avoir une densité spectrale de niveau X donné (en  $V^2/Hz$ ), il suffit d'adjoindre un coefficient multiplicatif de tension égal à  $\sqrt{X}$ .




source de tension commandée par un courant

Sous Pspice, cela se traduit par la netliste ci-contre.

La source de bruit est un sous-circuit, pour plus de facilité.

Dans l'exemple ci-contre, le coefficient multiplicatif est de  $10^{-7}$ , afin d'avoir une densité spectrale de  $10^{-14} V^2/Hz$ .

Pour satisfaire le fait d'avoir une source d'entrée pour réaliser une analyse AC, on ajoute une source factice sur un autre nœud.



```

source de bruit blanc
* fichier source.cir
.SUBCKT Vbruit N1 N2
Rn 1 0 {4*1.380649e-23*300}
Vn 1 0 DC=0
Hn N1 N2 Vn 1e-7
.ENDS


X1 1 0 Vbruit
.NOISE V(1) Vbidon
Vbidon 2 0 DC=0
Rbidon 2 0 1k

.AC DEC 10 0.1 10MEG
.probe
.end
    
```

Rappel :

L'affichage de NTOT(ONoise) indique le bruit total sur la sortie, ici V(1) par le paramètre de la directive .NOISE.

Dans notre cas, c'est créé uniquement par la résistance Rn, affectée du coefficient 1E-7.



#### Interprétation

La valeur affichée est de 10 f, soit  $10^{-14} V^2/Hz$  sur tout l'axe des fréquences : on a bien généré une source de bruit blanc parfaite, à la valeur désirée.

La valeur efficace de la tension de bruit, dans cette bande passante [0,1 Hz ; 10 MHz], est donnée par :

$$V_{\text{eff}}^2 = \int_{0,1}^{10M} \gamma(f) df = 10^{-14} [\ln 10M - \ln 0,1] = 10^{-14} [16,11 - (-2,3)] = 1,84 \cdot 10^{-12}, \text{ soit } V_{\text{eff}} = 1,36 \mu V.$$

### 3.b. Simuler un bruit blanc numérique

La solution la plus simple est de fabriquer, de façon externe, un fichier de points qui servira de paramètre pour une source PSpice PWL.

Remarque : cela a été réalisé dans l'article 69, et dans l'article 87.

Il s'agit donc d'une source temporelle, qui est connectée au nœud 1.

```
source PWL et excel
* fichier bruit_numerique.cir

.TRAN 1u 100m 0u 1u

Vbruit 1 0 PWL (FILE=alea.txt) AC=1
* un point toutes les 1 us
* 100ms d'enregistrement
Rbidon 1 0 10k


.probe
.end
```

Le fichier (généralisé par Excel, grâce à la fonction ALEA), comporte une coordonnée (temps, valeur) toutes les  $\mu\text{s}$ . Il y a 100 000 points, pour former un signal durant 100 ms.

On peut vérifier en zoomant sur le début de la simulation que le signal est formé de points toutes les  $\mu\text{s}$ .

Comme pour toute source PWL, entre chaque point, Pspice interpole par une droite.

L'affichage de  $\text{RMS}(V(1))$  donne 0,5 V efficace, ce qui parfaitement cohérent.




*extrait de la source PWL :  
toutes les  $\mu\text{s}$ , une valeur aléatoire entre -1 et +1*

Pour vérifier son profil de puissance, il faut afficher le spectre en tension par le menu FFT.

Vérifions la réponse en sinus cardinal.

Comme prévu, le premier zéro est à 1 MHz

Le bruit sera blanc jusqu'à 50 kHz



*Spectre, en tension, obtenu par FFT :  
une enveloppe en sinus cardinal qui passe par 0 à 1 MHz*

Remarque : comme indiqué dans l'article 72, il faut savoir paramétrer l'analyse .TRAN pour avoir une FFT satisfaisante.

## 4) Simuler le générateur de bruit rose complet


(netliste en **annexe 3** : generateur\_complet.cir)

### 4.a) En fréquentiel avec source bruit blanc

C'est la fusion des 2 essais précédents : 3a et 2b : la **source de bruit parfaite** (sous-circuit) est connectée au filtre.

- Dans un premier temps, on désactive le bruit en plaçant un coefficient = 0.  
On a alors en sortie une densité spectrale de tension de bruit quasi constante de  $220 \cdot 10^{-15} \text{ V}^2/\text{Hz}$  : c'est le bruit propre apporté par l'ensemble des composants électroniques. En consultant le fichier de sortie, on peut lire que c'est essentiellement la contribution de R9 (150 kΩ sur l'entrée - de l'A.Op) qui prédomine. (Dans cet essai, on suppose fiable le modèle en bruit du LF411).

- Dans un deuxième temps, on choisit une source de bruit de niveau  $10^{-8} \text{ V}^2/\text{Hz}$  (coefficient de  $10^{-4}$ ).



L'axe des ordonnées est en échelle logarithmique, pour mieux identifier la pente à  $-10 \text{ dB / décade}$ . Sans surprise, on retrouve le profil du filtre, puisque l'entrée est constante.


Vérifions le point à 10 kHz : en entrée, nous avons  $10^{-8} \text{ V}^2/\text{Hz}$ . L'atténuation en tension du filtre est de 0,125, donc 0,015625 en puissance.

La valeur attendue en sa sortie est donc  $10^{-8} \text{ V}^2/\text{Hz} \times 0,015625$ , soit  $1,56 \cdot 10^{-10}$ .

C'est ce qu'on retrouve :


X Values	9.995K
NTOT(ONOISE)	156.624p

Remarque : le même essai mais avec un niveau d'entrée plus faible (coefficient de  $2 \cdot 10^{-6}$ , donc  $4 \cdot 10^{-12} \text{ V}^2/\text{Hz}$ ) aboutit à une courbe translatée en basse fréquence, mais qui se stabilise à  $220 \cdot 10^{-15} \text{ V}^2/\text{Hz}$  en milieu de fréquence : c'est le bruit interne au montage qui subsiste.



*à faible niveau de bruit d'entrée, la réponse en sortie du filtre laisse paraître le bruit propre du montage*


Reprenons la simulation avec la source de bruit de niveau  $10^{-8} \text{ V}^2/\text{Hz}$ , mais présentons la sortie sur une échelle linéaire :



Sur cette représentation, on peut voir aisément que la **densité spectrale** de sortie est confondue avec la loi  $\frac{1,4 \cdot 10^{-6}}{f}$ .

#### 4.b) En temporel avec source bruit numérique

C'est la fusion des 2 essais précédents : 3b et 2b. La source de bruit est réalisée par la source PWL qui est une suite de points entre -1 +1 toutes les  $\mu\text{s}$ .



#### Interprétation

La simulation ne faisant de 100 ms, il est impossible de voir l'action du filtre pour des fréquences très faibles. Néanmoins, on identifie une variation lente qui domine, sur laquelle des fluctuations plus rapides se superposent.

Et pour ajouter de la beauté et de l'humour, nous pouvons voir que la trace est rose...


#### Conclusion

Cette étude a montré le principe des sources de bruit blanc, ainsi que le filtre pour générer du bruit rose.

Sur ce filtre, pour respecter la pente à  $-10 \text{ dB / décade}$ , il faut des valeurs numériques particulières au réseau RC. Il n'existe pas d'équations à résoudre pour déterminer les valeurs individuelles de R1, C1, R2, C2... etc. La simulation analogique permet de valider un choix de valeurs, de corriger pour affiner (c'est ce qui a été fait pour le schéma simulé, un peu différent de celui d'origine, le PNG11), et de mieux juger l'étendue de fréquence exploitable.

Le générateur complet est suffisamment simple pour être simulé avec la version d'éval de Pspice.

**Annexe 1** Schéma du générateur PNG11



**Commentaires :**

*J'ai exploité ce matériel, il y a... très longtemps. A l'époque (après un repas à base de dinosaure), j'ai eu la bonne idée de photocopier ce schéma, issu de la documentation constructeur. J'avoue que je n'ai pas souvenir du nom de ce constructeur.*

Ce schéma est très pédagogique et clairement explicite. On reconnaît :

- un circuit qui génère une horloge, à base d'une structure exploitant les XOR et un circuit RC. La fréquence obtenue n'est pas précise, mais pour cette application, c'est justifié. Il est indiqué 2 MHz.
- le PRPG, à base de 3 CD4006 (18 bascules D chacun). Il est indiqué qu'on exploite 39 étages. Un XOR (CD4070) permet un rebouclage pour obtenir la séquence la plus longue soit  $N = 2^{39}-1 \approx 550$  milliards... Avec une fréquence d'horloge 2 MHz, l'écart entre 2 raies est de 3,6  $\mu$ Hz, et le bruit est blanc jusqu'à 100 kHz. La période de répétition de la succession de 0 et 1 est de 550 milliards x 0,5  $\mu$ s, soit plus de 3 jours...
- filtre passe bas (à - 10 dB/décade) permettant le bruit rose.
- filtre passe bande qui permet d'avoir du bruit à bande passante limité.
- un filtre de sortie qui borne à 100 kHz le spectre de sortie.

Sur la face avant de cet appareil, un sélecteur permettait de choisir un bruit parmi 3, et on disposait d'un réglage de niveau, indiqué par LEVEL sur le schéma.

Remarque : dans nos simulations, quelques valeurs numériques ont été modifiées pour améliorer la réponse harmonique.

**Annexe 2** Netliste Pspice paragraphe 2a : Le filtre a – 10 dB/décade

```

filtre à - 10 dB/décade
* fichier filtre.cir
* - 10 dB/dec de 10 Hz à 100 kHz

.lib eval.lib

Vplus Vplus 0 DC=12V
Vmoins Vmoins 0 DC=-12V

*Ve 1 0 AC=1 ; pour réponse harmonique
*.AC DEC 100 0.1 10MEG

V1 0 a sin (0 1 100)
V2 a b sin (0 1 1k)
V3 b 1 sin (0 1 10k)
.TRAN 1u 100m 10m 1u
.IC V(out)=0

R0 1 Out 10k
R1 Out 2 10k
C1 2 0 2.2u
R2 Out 3 4.7k
C2 3 0 1u

R3 Out 4 2.2k
C3 4 0 470n
R4 Out 5 1k
C4 5 0 220n
R5 Out 6 470
C5 6 0 100n
R6 Out 7 220
C6 7 0 47n
R7 Out 8 82
C7 8 0 22n
R8 Out 9 33
C8 9 0 10n
C9 Out 0 4.7n

CL Out eplus 100u
R9 eplus 0 150k
X2 eplus emoins Vplus Vmoins 100 LF411
Ra 100 Si 18k
Rb Si 0 2.2k
R10 Si Emoins 150k

.probe
.end

```

**Annexe 3** Netliste Pspice paragraphe 4 : générateur de bruit rose complet

```

montage generateur de bruit
* fichier generateur_complet

.lib eval.lib

.SUBCKT Vbruit N1 N2
Rn 1 0 {4*1.380649e-23*300}
Vn 1 0 DC=0
Hn N1 N2 Vn 1e-4 ; 0 ; 2e-6
.ENDS

*Ve 1 0 DC=0 AC=1
*.AC DEC 1000 0.1 10MEG

*X1 1 0 Vbruit
.noise V(100) Vbidon 30
Vbidon 20 0 DC=0
Rbidon 20 0 1k

Vbruit 1 0 PWL (FILE=alea.txt) AC=1
Rbidon2 1 0 10k
.TRAN 1u 100m 0u 1u

Vplus Vplus 0 DC=12V
Vmoins Vmoins 0 DC=-12V

R0 1 Out 10k
R1 Out 2 10k
C1 2 0 2.2u
R2 Out 3 4.7k
C2 3 0 1u
R3 Out 4 2.2k
C3 4 0 470n
R4 Out 5 1k
C4 5 0 220n
R5 Out 6 470
C5 6 0 100n
R6 Out 7 220
C6 7 0 47n
R7 Out 8 82
C7 8 0 22n
R8 Out 9 33
C8 9 0 10n
C9 Out 0 4.7n

CL Out eplus 100u
R9 eplus 0 150k
X2 eplus emoins Vplus Vmoins 100 LF411
Ra 100 Si 18k
Rb Si 0 2.2k
R10 Si Emoins 150k

.IC V(out)=0
.probe
.end

```

articles 1 à 43 : sur le livre

**Tableau récapitulatif des articles PDF disponibles sur ce site**

n°	titre	lien présentation	lien direct article
	Guide d'installation et d'emploi simplifié	<a href="#">présentation</a>	<a href="#">document PDF</a>
44	Exemples basiques et des exercices...	<a href="#">présentation</a>	<a href="#">document PDF</a>
45	Un exemple de circuit passif	<a href="#">présentation</a>	<a href="#">document PDF</a>
46	Un oscillateur Colpitts	<a href="#">présentation</a>	<a href="#">document PDF</a>
47	Compensation en fréquence des amplificateurs opérationnels	<a href="#">présentation</a>	<a href="#">document PDF</a>
48	Un amplificateur à transistors bipolaires	<a href="#">présentation</a>	<a href="#">document PDF</a>
49	Une bascule D Flip Flop CMOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
50	Une porte XOR à transistors MOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
51	Un VCO à 12 transistors MOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
52	Une PLL à moins de 20 transistors MOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
53	Un oscillateur à résistance négative	<a href="#">présentation</a>	<a href="#">document PDF</a>
54	Une charge électronique	<a href="#">présentation</a>	<a href="#">document PDF</a>
55	Un amplificateur en classe C	<a href="#">présentation</a>	<a href="#">document PDF</a>
56	Le monostable 74 123	<a href="#">présentation</a>	<a href="#">document PDF</a>
57	Un amplificateur en classe D	<a href="#">présentation</a>	<a href="#">document PDF</a>
58	Le transformateur en linéaire	<a href="#">présentation</a>	<a href="#">document PDF</a>
59	La loi d'ohm thermique	<a href="#">présentation</a>	<a href="#">document PDF</a>
60	Le transformateur en non linéaire	<a href="#">présentation</a>	<a href="#">document PDF</a>
61	Robustesse d'un oscillateur en anneau	<a href="#">présentation</a>	<a href="#">document PDF</a>
62	Une alimentation stabilisée	<a href="#">présentation</a>	<a href="#">document PDF</a>
63	Modélisation d'un haut-parleur	<a href="#">présentation</a>	<a href="#">document PDF</a>
64	Un synthétiseur de fréquence	<a href="#">présentation</a>	<a href="#">document PDF</a>
65	Un ampli audio de Sparkfun	<a href="#">présentation</a>	<a href="#">document PDF</a>
66	Simulation logique et analogique	<a href="#">présentation</a>	<a href="#">document PDF</a>
67	Un oscillateur à relaxation	<a href="#">présentation</a>	<a href="#">document PDF</a>
68	Lecteur de TAG RFID 125 kHz	<a href="#">présentation</a>	<a href="#">document PDF</a>
69	Diagramme de l'œil avec Pspice	<a href="#">présentation</a>	<a href="#">document PDF</a>
70	Un amplificateur hautes fréquences	<a href="#">présentation</a>	<a href="#">document PDF</a>
71	Une bizarrerie enfin expliquée...	<a href="#">présentation</a>	<a href="#">document PDF</a>
72	Comprendre le paramétrage de la FFT	<a href="#">présentation</a>	<a href="#">document PDF</a>
73	La relation de Bennett	<a href="#">présentation</a>	<a href="#">document PDF</a>
74	Simuler un circuit à plus de 20 transistors avec PSpice Eval	<a href="#">présentation</a>	<a href="#">document PDF</a>
75	Une horloge biphase sans recouvrement	<a href="#">présentation</a>	<a href="#">document PDF</a>
76	Quelques simulations sur la diode	<a href="#">présentation</a>	<a href="#">document PDF</a>
77	Un ampli classe A, avec transformateur de sortie	<a href="#">présentation</a>	<a href="#">document PDF</a>
78	Des stimuli pour PSpice	<a href="#">présentation</a>	<a href="#">document PDF</a>
79	Simuler le TL431 : zener ajustable	<a href="#">présentation</a>	<a href="#">document PDF</a>
80	Un ADC flash	<a href="#">présentation</a>	<a href="#">document PDF</a>
81	Une chaîne d'acquisition : S&H, ADC, DAC	<a href="#">présentation</a>	<a href="#">document PDF</a>
82	Un amplificateur 50 MHz	<a href="#">présentation</a>	<a href="#">document PDF</a>
83	Un dérivateur non inverseur	<a href="#">présentation</a>	<a href="#">document PDF</a>
84	Un amplificateur bipolaire avec push pull CMOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
85	Rôle des répéteurs logiques dans un circuit intégré	<a href="#">présentation</a>	<a href="#">document PDF</a>
86	Un driver logique CMOS pour charge 50 ohms	<a href="#">présentation</a>	<a href="#">document PDF</a>
87	Des triggers de Schmitt et des applications	<a href="#">présentation</a>	<a href="#">document PDF</a>
88	Un filtre gaussien analogique	<a href="#">présentation</a>	<a href="#">document PDF</a>
89	Un générateur de bruit rose	<a href="#">présentation</a>	<a href="#">document PDF</a>

90	Un anémomètre à fil chaud : simulation comportementale	<a href="#">présentation</a>	<a href="#">document PDF</a>
91	Un oscillateur à pont de Wien stabilisé par CTN	<a href="#">présentation</a>	<a href="#">document PDF</a>
92	L'emballement thermique d'une diode	<a href="#">présentation</a>	<a href="#">document PDF</a>
93	Les puissances dans un amplificateur	<a href="#">présentation</a>	<a href="#">document PDF</a>
94	Asservissement de puissance dans une résistance	<a href="#">présentation</a>	<a href="#">document PDF</a>
95	Asservissement de la puissance émise par une antenne radio	<a href="#">présentation</a>	<a href="#">document PDF</a>
96	Un driver de LED de puissance	<a href="#">présentation</a>	<a href="#">document PDF</a>
97	Exploiter Pspice pour simuler des filtres numériques	<a href="#">présentation</a>	<a href="#">document PDF</a>
98	Un filtre en cosinus surélevé avec Pspice	<a href="#">présentation</a>	<a href="#">document PDF</a>
99	Effet de la température sur un amplificateur en classe A	<a href="#">présentation</a>	<a href="#">document PDF</a>
100	Un amplificateur à transistors JFET et bipolaires	<a href="#">présentation</a>	<a href="#">document PDF</a>
<b>Supplément, hors article :</b>			
<b>mon cours « Electronique pour les communications numériques », <a href="#">polycopié couleur 201 pages en pdf</a></b>			

[retour à l'écran d'accueil de ce site](#)